

BIOMETRICS INFORMATION

(You're 95% likely to need this information)

PAMPHLET NO. # 38

DATE: June 18, 1992

SUBJECT: Plotting Error Bars with SAS/GRAPH

In our recent "Introduction to SAS/GRAPH Workshop", a program using annotated data sets for plotting error bars was discussed. This pamphlet describes a simpler method for performing the same task.

Assume we have a data set containing height measurements of seedlings treated with four different fertilizers. We want to compute the mean seedling heights for the four groups, and plot bars of one standard deviation wide around the means. Recall that a one standard deviation bar is a vertical line through the mean, extending beyond the mean by plus and minus one standard deviation. The desired graph can be obtained with SAS/GRAPH by selecting the appropriate options in the SYMBOL statement.

The SYMBOL statement defines the type of line to be plotted. Options in the SYMBOL statement that are frequently used are:

C = to specify the color of the plotted lines;

L = to specify the style of the plotted lines;

I = to specify the interpolation method to be used to plot the lines.

See *SAS/GRAPH User's Guide, Version 6.03* for a detailed description of these and other options. I = JOIN and I = SPLINE are the two options commonly chosen for interpolation. For data sets containing multiple y-values for each level of the x-variable, error bar plots of various forms can be obtained by selecting the option I = STDkxxx. Bars of 1, 2, or 3 standard deviations can be selected by using the appropriate value for k. The default when k is omitted is two standard deviations. For example:

I = STD1 would produce bars of one standard deviation.

The letters following k can be M, P, J, B, T, or any combination of these letters except B with T.

I = STDkM would produce bars of standard errors
(standard error = standard deviation/ \sqrt{n}).

I = STDkP would produce bars of standard deviations where standard deviations are computed using a pooled estimate, as in a one-way ANOVA method.


I = STDkJ would produce bars of standard deviations with the means jointed by a line from bar to bar.

I = STDkB would produce bars in the form of a rectangular box rather than lines.


I = STDkT would produce bars with top and bottom added to each line.

The following are error bar plots from various `I =` options.


1a) `I = STD1J`




1b) `I = STD1P`



1c) `I = STD1MT`



1d) `I = STD1B`



The SAS code for graph (1c) is given in the Appendix.

Note that the range of the y-axis is defined in the `AXIS1` statement with the `ORDER =` option. If the range is not specified then SAS will choose a range that will accommodate all data points. Often this scale is too large, resulting in the bars being too small for comparison. The x-variable can be either numeric or character. For character variables, the order of the values can be defined in the `AXIS` statement with the `ORDER =` option. For example, the option `ORDER = ('Fert-3' 'Fert-1' 'Fert-2' 'Fert-4')` would produce a graph with the x-values in the defined order.

This method for plotting error bars is easy to use, but is not versatile. For instance, you cannot choose the plotting symbol for the means. Moreover, only bars of standard deviation or standard error are possible with this method. When the sample size is less than 30 and the Normal approximation is not valid, it may be desirable to construct confidence limits around the means. In this case, the method described here will not be sufficient and the annotated data sets method described in Sit (1992) must be used.

References

- SAS Institute Inc. 1988. SAS/GRAPH user's guide, version 6.03 edition. SAS Institute Inc., Cary, N.C.
- Sit, V. 1992. Introduction to SAS/GRAPH notes. Biometrics, Res. Br., Min. For., Victoria, B.C.

CONTACT: Vera Sit
356-0435

Appendix: The SAS code for graph (1c)

```

options ls=78 nodate;

data trees;                                /* input height data */
  if _n_=1 then tree=1;
  do f=1 to 4;
    fert = 'Fert-'||put(f,1.0);          /* create values with */
    input ht@@;                          /* Fert- as prefix */
    output;
  end;
  tree+1;
cards;
  40.6 38.8 24.7 27.8
  28.1 56.0 21.7 37.5
  42.3 33.1 38.0 25.9
  35.8 31.9 27.1 41.5
  28.1 54.0 22.5 26.6
  36.5 32.2 39.1 25.9
  42.8 54.9 26.1 27.1
  35.2 27.9 28.7 41.6
  38.4 57.6 35.7 35.3
  36.7 53.6 37.2 26.7
;

goptions reset=all device=ps border chartype=5 vsize=4.6
          hsize=7.0 vpos=46 hpos=70 noprompt csymbol=black;

title ls=1 h=2.00 f=none 'One Standard Error Bars';
title2 ls=0.5 h=2.0 'with Ends';
footnote ' ';

symbol1 v=none i=stdlmt mode=include;

axis1 order=(20 to 60 by 5) label=(a=90 h=1.5 'Height (mm)')
      offset=(0) value=(h=1.3) minor=none;
axis2 label=(h=1.5 'Fertilizer' ) value=(h=1.3) offset=(1cm) ;

proc gplot data=pamp.trees;
  plot ht*fert / vaxis=axis1 haxis=axis2;
run;

```